
1 Scenario
1.1 Simple interaction between two variables. 

This example shows how one variable can be filled based on the value of another variable. This can be used in various circumstances. The implementation details are listed in the chapters 3.2 Basics and 3.3 Simple interaction between variables – implementation.

· Next year: You have a variable A containing the actual year and you would like to fill variable B for the next year automatically based on A.

· Reference versions: You have a variable A containing your current plan version and you would like to fill variable B with the corresponding reference version.

· Global variables: From a logical point of view you have the same variables in several planning areas e.g. the actual year. Now you would like to define one ‘leading’ area containing the variable that should automatically set/adjust the corresponding variables in the other planning areas.

1.2 Use Variables of type attribute in the WEB or in the planning folders

Due to technical reasons you can’t use variables of type attributes in the web or in the planning folders. A workaround is explained in chapter 3.4 Using variables of type attribute.

1.3 Selection of an interval

This example shows how you can fill a variable with an interval by specifying the ‘from’ and the ‘to’ separately, e.g. with two drop down boxes in a WEB Application. This is useful when you need one variable with an interval (e.g. for data selection in a planning package) as well as the upper and lower values of the interval (e.g. for the parameters of a planning function). Please see chapter 3.5 Selection of an interval.

1.4 One variable restricts the selection of another variable.

You work in the WEB and you have a variable A containing the material group. Now variable B should contain all materials belonging to this group.

1.5 BW Query uses SEM-BPS/BW-BPS variable for default value

During your planning session you have restricted a variable to a single value. Now you like to execute a BW query that should use this restriction as default. Please see chapter 3.7 BW Query uses SEM-BPS/BW-BPS variable for default value for more details.
2 Introduction

Before starting there are some things that have to be taken into consideration when working with the examples.

When the system sets or reads the values of several variables in one step (e.g. when pressing the refresh button) no particular sequence of processing is guaranteed. This could lead to problems when you work with ‘dependent’ variables as shown in the following example. You can force the system to set and refresh a variable value after each change by pressing the refresh button in the planning folder each time or, for variables in the WIB, by setting the event-attribute to true.

Example: You allow the choice of the material group and the material simultaneously with the consequence that you can enter an invalid combination. Hence you have to specify a rule that determines which variable is the dependent and which is the leading one in order to determine the result values.

The variables are often part of the selection. Please consider that for one execution of a planning function the value is normally requested several times. This implies that it doesn’t make sense to send pop-ups without additional logic. As a consequence of this you should buffer the selection results in the variable to avoid a certain overhead. 
During processing using the default settings, the WIB buffers the values of a variable internally and will not change the set of variable values that are displayed after the initial execution of a WEB planning application. This default can be changed by specifying a variable that should trigger a re-read of a second variable. (When using SEM3.1-3.5 you need to add a record to table upc_dark2 with param= WEB_REFRESH_ON_CHANGE and value=X.) 

Example: The variable product delivers the values 1000 and 1001 on the execution of the web application. As this variable is of type exit it’s values can change during runtime to 2000 and 2001 (e.g. as a reaction to the change of variable for product line). The drop-down box on the WEB page will still show the values 1000 and 1001 unless you have specified that a change of product line should re-read the values of the drop-down box.
To understand the steps described in the following chapters it is assumed that you are familiar with the ABAP development transactions (e.g. SE80 or SE37) and that you know how to create and activate new function modules.
3 The Step By Step Solution  

3.1 Introduction

All examples are based on the delivered Info Cube 0SEMBPSPF and the planning area 4PERF001.

In the drop down boxes you see the technical name and the text of the entries. This will allow you to uniquely identify the right entries when working in other languages than English.

The display of the technical names can be activated via the button [image: image3.bmp] in the SAPGui menu bar that brings up the following pop-up. By checking the box ‘Show keys in all dropdown lists’ you activate the display of the additional information in the drop down lists.

3.2 Basics

Most of the examples need the capability to read values of another variable. Hence this topic is separated and shown in the first step. The function module introduced reads the values of characteristic and attribute variables. It provides …

· all values of a variable 

(table eto_varsel_all).

· the restricted value 

(table eto_varsel).

· the name of the characteristic(s) (table eto_chanm).

· The capability of returning the buffered values of the last call (i_buffer). If you call the function module with a combination of planning area (i_area) and variable name (i_varname) that differs from the last call, the buffer is invalidated and the values are newly determined. Depending on your scenario it could be helpful to further improve this logic on your own. 

The table below shows the interface definition of this function module with the name ‘Z_VARIABLE_GET_DETAIL’.

	1. Create the function module z_variable_get_detail. The complete Source code is listed in the attachments of this document.
(See chapter  4.1 Function module Z_VARIABLE_GET_DETAIL).



	FUNCTION z_variable_get_detail.

*”-------------------------------------------------

*”*”Local interface:

*”  IMPORTING

*”     VALUE(I_AREA) TYPE  UPC_VAR-AREA

*”     VALUE(I_VARIABLE) TYPE  UPC_VAR-VAR

*”     VALUE(I_BUFFER) TYPE  BOOLE-BOOLE OPTIONAL

*”  EXPORTING

*”     VALUE(E_SUBRC) TYPE  SY-SUBRC

*”     VALUE(ES_RETURN) TYPE  BAPIRET2 

*”     VALUE(E_TYPE) TYPE  UPC_VAR-VARTYPE

*”     REFERENCE(ETO_VARSEL_ALL) TYPE  UPC_YTO_CHARSEL

*”     REFERENCE (ETO_VARSEL) TYPE  UPC_YTO_CHARSEL

*”     REFERENCE (ETO_CHANM) TYPE  UPC_YTO_CHA

*”-------------------------------------------------




3.3 Simple interaction between variables – implementation

3.3.1 Determine next year

	1. The following steps have to be done in the delivered planning area 4PERF001 – ‘Basic technical performance’.

Create the variable SOURCE of variable type characteristic. 

	2. Chose the characteristic ‘Fiscal year’ and enter a value, e.g. 2002. Do not check the box ‘Restriction of Values Required by User’.

	3. Create the variable DEST of variable type characteristic.

	4. Chose as details as shown in the screenshot. It is important to select the replacement type User Exit.

	5. Navigate to transaction SE37 and create the function module Z_SIMPLE_RELATION by pressing the tools button at the right side of the function module name.

	6. Enter the import and export parameter as shown. This interface definition is given by SEM-BPS/BW-BPS.
	FUNCTION Z_SIMPLE_RELATION.

*"-------------------------------------------------

*"*"Lokale Schnittstelle:

*"  IMPORTING

*"     VALUE(I_AREA) TYPE  UPC_Y_AREA

*"     VALUE(I_VARIABLE) TYPE  UPC_Y_VARIABLE

*"     VALUE(I_CHANM) TYPE  UPC_Y_CHANM OPTIONAL

*"     VALUE(ITO_CHANM) TYPE  UPC_YTO_CHA

*"  EXPORTING

*"     REFERENCE(ETO_CHARSEL) TYPE  UPC_YTO_CHARSEL

*"-------------------------------------------------



	7. Add the Source code mentioned in the attachment 4.2 Function module Z_SIMPLE_RELATION. It contains two parts: the first one reads the value of the source variable, and the second one introduced by the comment ‘Example 1’ show how to derive the new values. 
	( Attachment: Function module Z_SIMPLE_RELATION

	8. Save and activate the function module
	

	9. Go back to SEM-BPS/BW-BPS using the green back arrow of the SAP-Gui and save the newly created customizing.
	

	10. Test the results in the transaction BPS0: call up the set variable screen …

Pop-up shows:

	11. … and use F4-Help to set the source value to e.g. year 2002.

	12. Now set the variable DEST using the F4-Help and you will see, that it shows the value of 2003. This is the source value of 2002 incremented by one year.

	13. .Now we have our first working example.

To learn more about the features of the exits follow the next steps, which are grouped into several parts.
	

	14. Hints Part 1
When changing the SOURCE variable you will see that the result will not change.
	

	15. The reason for this is, that the values are buffered in the function module Z_VARIABLE_GET_DETAIL by default. 

By changing the value from ‘X’ to ‘ ‘ you can avoid this.

The reason for this default is performance related: It is faster to return the buffered value directly. 

Please remember that the content of the drop down boxes in the WEB will be defined during the start and not reread during the planning session. 

This has the effect that the values of the DEST variable will not change in a drop-down box but only in the planning levels etc. .
	FUNCTION z_simple_relation.

[… other lines of Source code…]

l_buffer_call TYPE boole-boole VALUE 'X'.

[… other lines of Source code…]



	16. Hints Part 2
When entering two or three values in the SOURCE variable and checking the flag for user restriction, the function module z_simple_relation will stop with an information message saying, that only one value is allowed. The reason is that the function module reads by default all values of a variable and it doesn’t care about the restriction.

	17. To change this, replace the blank with an ‘X’.
	[… other lines of Source code…]

l_use_restricted_values TYPE boole-boole VALUE 'X',

[… other lines of Source code…]




Now we have seen how the first example can be implemented. But how can we use the example to solve the other scenarios mentioned? 

3.3.2 Reference version

For the scenario where you would like to determine the value of a variable (reference version) dependent on the value of another variable (current plan version), you have to implement your own logic between the comment lines containing ‘logic start’ and ‘logic end’. As the rules are dependent on your specific set of requirements we can’t make any suggestions here.

3.3.3 Global variables

From a logical point of view you have the same variables in several planning areas e.g. the actual year. Now you would like to define one ‘leading’ area containing the variable that should automatically set/adjust the corresponding variables in the other planning areas.

If you would like to refer to the values of a reference variable (example 3) you simply have to change the constants containing the name of the variable and the area: in all areas where you implement the DEST variable the constants should point to your SOURCE variable.

In addition, you delete all coding between the ‘logic start’ and ‘logic end’ lines as there is no additional logic to be implemented.

* ------------------------------- logic start

* nothing to do

* ------------------------------- logic end

3.4 Using variables of type attribute

The basic idea of the solution is to use a variable SOURCE of type characteristic on the front-end side and to transfer the selected value to a variable DEST of type attribute on the back-end side. This attribute variable could then be used to restrict the selection. 

As we are again merely transferring values from one variable to another we can make use of the implementation shown above. To do so follow these steps: 

· Choose the example of the global variables

· Change the variable type of the variable DEST from characteristic to attribute.

The implementation now works as follows: Add the variable DEST into the attribute selection of the planning level and refer to the SOURCE variable. As this variable has the normal type characteristic it can be used in the planning folders and in the WEB.

A change of the source now influences via our implemented logic directly the values of the attribute variable.

3.5 Selection of an interval

This problem is solved by creating two source variables SFROM and STO containing the ‘from’ and the ‘to’ values as single values and a variable DEST, that will read the values of the two sources and compile them in an interval. As there are several code changes necessary, the complete Source code has been added to the appendix (4.3 Selection of an interval). 

3.6 One variable restricts the selection of another variable

Let us assume that we have a characteristic 0bps_produ (products) containing the master data attribute 0bps_prodl (product line). When selecting one product line the variable for the products should contain only the set of values that corresponds to the attribute selection.

We assume further that the variable for the product line has the name SOURCE and the variable for the product has the name DEST in order to be able to reuse the source code of the previous examples.

Using the example Z_SIMPLE_RELATION please replace the Source code that determines the logic to find the next year by the Source code that determines the attribute values.
3.6.1 Source code: Next year – to be deleted

* is there a next year?

  l_next_year = ls_varsel-low.

  IF l_next_year = 9999.

    MESSAGE i003(upc) WITH '9999 +1' '0FISCYEAR'.

*   Value &1 is not permitted for characteristic &2

    EXIT.

  ENDIF.

* find next year

  l_next_year = l_next_year + '0001'.

  ls_varsel-low = l_next_year.

* add to result table

  APPEND ls_varsel TO eto_charsel.
3.6.2 Source code: Attribute values

* [lines 89ff.]

* This example will only work with local BW installation!

* determine the bps_produ belonging to the bps_prodl

  SELECT * FROM /bi0/mbps_produ INTO TABLE lt_chavl
    WHERE bps_prodl = ls_varsel-low.

  if sy-subrc <> 0.

    MESSAGE e026(upc) WITH ls_varsel-low '0BPS_PRODL'.

*   attribute value not found

    exit.

  endif.

  ls_varsel-seqno = '0000'.

  ls_varsel-chanm = '0BPS_PRODU'.

  LOOP AT lt_chavl INTO ls_chavl.

    ls_varsel-seqno = ls_varsel-seqno + 1.

    ls_varsel-low = ls_chavl-bps_produ.

    APPEND ls_varsel TO eto_charsel.

  ENDLOOP.

3.7 BW Query uses SEM-BPS/BW-BPS variable for default value

The implementation consists of two parts: 

1. Defining a SEM-BPS/BW-BPS variable with the attribute ‘restriction of values required by the enduser’ set. In the following description the name of this variable is ‘RESTRVAL’.

2. Create a Bex variable of type customer exit. Within this customer exit you determine the restricted value of the SEM-BPS/BW-BPS variable and set the default value of the Bex variable accordingly.


	1. Start the BW query designer and create a query where we use e.g. the country (0BPS_CNTRY). Our example is based on the InfoCube 0SEMBPSPF which belongs to the planning area 4PERF001 used in the examples above.

	2. Save the query and execute the query in WEB by pressing the [image: image16.bmp] button.

	3. Now should see a similar result depending on the data in your InfoCube …

	4. Create a variable for SEM-BPS/BW-BPS Country (0BPS_COUNTRY) with the settings as shown on the following screen shots.

	5. Change settings according to the values shown on the right. Use ‘Next’-Button to navigate to next screen.

	6. Change settings according to the values shown on the right. Use ‘Next’-Button to navigate to next screen.

	7. Press the ‘Finish-Button.

	8. Move the variable via drag & drop to the characteristic SEM-BPS/BW-BPS Country in the area of the free characteristics and save the query.

	9. Now create a variable with the name ‘RESTRVAL’ in the planning framework (transaction BPS0) according to the picture on the right. For details on how to create such a variable please refer to previous chapters or the SEM-BPS/BW-BPS documention.

	10. Press the button ‘Set Variables’. 
	[image: image25.png]\Business Planning and Simulation

Gloss navigaton | [ | | Planing rofi || | Giobatplanning sequences | | | setvariabies






	11. Choose one country e.g. CH and save the change made.

	12. Start the transaction SE38 and change the program with the name ZXRSRU01.

	13.  Add the coding shown to the program.
	

	*&---------------------------------------------------------------------*

*&  Include           ZXRSRU01                                         *

*&---------------------------------------------------------------------*

DATA: l_subrc LIKE sy-subrc,

      ls_return LIKE bapiret2,

      lto_varsel TYPE upc_yto_charsel,

      ls_varsel TYPE upc_ys_charsel,

      l_s_range TYPE rsr_s_rangesid.

CONSTANTS:

* adjust the varnames to your needs:

      l_varname_sem LIKE upc_var-var VALUE 'RESTRVAL',

      l_varname_bw  TYPE rszglobv-vnam VALUE 'SEMCNTRY',

* specify the area where the variable can be found

      l_area    TYPE upc_y_area  VALUE '4PERF001',

      l_buffer  TYPE c VALUE ' '.

IF i_vnam EQ l_varname_bw AND i_step = 1.

  CALL FUNCTION 'Z_VARIABLE_GET_DETAIL'

    EXPORTING

      i_area     = l_area

      i_variable = l_varname_sem

      i_buffer   = l_buffer

    IMPORTING

      e_subrc    = l_subrc

      es_return  = ls_return

      eto_varsel = lto_varsel.

  IF l_subrc <> 0.

    CALL FUNCTION 'RRMS_MESSAGE_HANDLING'

      EXPORTING

        i_class  = 'UPF'

        i_type   = 'E'

        i_number = '001'

        i_msgv1  = 'Cannot fill Bex variable'

        i_msgv2  = l_varname_bw

        i_msgv3  = 'based on Sem variable'

        i_msgv4  = l_varname_sem.

*   MESSAGE i149(upc_fw) WITH l_varname.

*   Value of variable &1 cannot be determined

    CALL FUNCTION 'RRMS_MESSAGE_HANDLING'

      EXPORTING

        i_class  = 'UPC_FW'

        i_type   = 'E'

        i_number = '149'

        i_msgv1  = l_varname_sem.

    EXIT.

  ENDIF.

* read first line of selection result

  READ TABLE lto_varsel INTO ls_varsel INDEX 1.

  IF sy-subrc <> 0.

*   MESSAGE i147(upc_fw) WITH l_varname_sem.

*   Variable &1 does not contain any values

    CALL FUNCTION 'RRMS_MESSAGE_HANDLING'

      EXPORTING

        i_class  = 'UPC_FW'

        i_type   = 'E'

        i_number = '147'

        i_msgv1  = l_varname_sem.

    EXIT.

  ENDIF.

* add values to return table

  l_s_range-sign   = ls_varsel-sign.

  l_s_range-opt    = ls_varsel-opt.

  l_s_range-low    = ls_varsel-low.

  l_s_range-high   = ls_varsel-high.

  APPEND l_s_range TO e_t_range.

ENDIF.

	14.  Start the transaction CMOD and choose the project RSR00001. Aktivate the enhancement by pressing the [image: image28.png]


 button.

	15.  Now we are ready to test the results: When performing your query the system should come up with a variable screen. The default value and the restricted value in SEM-BPS/BW-BPS should be the same, in our example the country ‘CH’.

4 Appendix (Sample ABAP Source code)  

4.1 Function module Z_VARIABLE_GET_DETAIL

FUNCTION z_variable_get_detail.

*"----------------------------------------------------------------------

*"*"Lokale Schnittstelle:

*"  IMPORTING

*"     REFERENCE(I_AREA) TYPE  UPC_VAR-AREA

*"     REFERENCE(I_VARIABLE) TYPE  UPC_VAR-VAR

*"     REFERENCE(I_BUFFER) TYPE  BOOLE-BOOLE OPTIONAL

*"  EXPORTING

*"     VALUE(E_SUBRC) TYPE  SY-SUBRC

*"     VALUE(ES_RETURN) TYPE  BAPIRET2

*"     VALUE(E_TYPE) TYPE  UPC_VAR-VARTYPE

*"     REFERENCE(ETO_VARSEL_ALL) TYPE  UPC_YTO_CHARSEL

*"     REFERENCE(ETO_VARSEL) TYPE  UPC_YTO_CHARSEL

*"     REFERENCE(ETO_CHANM) TYPE  UPC_YTO_CHA

*"----------------------------------------------------------------------

  TYPES:

    BEGIN OF ys_buffer,

      area          TYPE upc_var-area,

      variable      TYPE upc_var-var,

      subrc         LIKE sy-subrc,

      s_return      LIKE bapiret2,

      type          LIKE upc_var-vartype,

      to_varsel_all TYPE upc_yto_charsel,

      to_varsel     TYPE upc_yto_charsel,

      to_chanm      TYPE upc_yto_cha,

    END OF ys_buffer.

  STATICS:

    st_buffer   TYPE SORTED TABLE OF ys_buffer

                     WITH UNIQUE KEY area variable.

  DATA:

    ls_buffer   TYPE ys_buffer,

    lr_variable TYPE REF TO cl_sem_variable.

* Check buffer

  READ TABLE st_buffer INTO ls_buffer WITH TABLE KEY

    area     = i_area

    variable = i_variable.

* Found :-)

  IF sy-subrc = 0 AND i_buffer = 'X'.

    e_subrc        = ls_buffer-subrc.

    es_return      = ls_buffer-s_return.

    e_type         = ls_buffer-type.

    eto_varsel_all = ls_buffer-to_varsel_all.

    eto_varsel     = ls_buffer-to_varsel.

    eto_chanm      = ls_buffer-to_chanm.

    EXIT.

  ENDIF.

* Not found :-(

  CLEAR: e_subrc, es_return, e_type, eto_chanm, eto_varsel_all, eto_varsel.
    CALL METHOD cl_sem_variable=>get_instance

      EXPORTING

        i_area       = i_area

        i_variable   = i_variable

      RECEIVING

        rr_variable  = lr_variable

      EXCEPTIONS

        not_existing = 1

        OTHERS       = 2.

    IF sy-subrc <> 0.

      e_subrc = 4.

      CALL FUNCTION 'BALW_BAPIRETURN_GET2'

        EXPORTING

          type   = sy-msgty

          cl     = sy-msgid

          number = sy-msgno

          par1   = sy-msgv1

          par2   = sy-msgv2

          par3   = sy-msgv3

          par4   = sy-msgv4

        IMPORTING

          return = es_return.

      EXIT.

    ENDIF.

* read details of variable

    CALL METHOD lr_variable->get_attributes

      IMPORTING

        e_type    = e_type

        eto_chanm = eto_chanm.

* read restricted values

    CALL METHOD lr_variable->get_value

      EXPORTING

        i_user     = sy-uname

        i_restrict = 'X'

      RECEIVING

        rto_value  = eto_varsel

      EXCEPTIONS

        error      = 1

        OTHERS     = 2.

    IF sy-subrc <> 0.

      e_subrc = 4.

      CALL FUNCTION 'BALW_BAPIRETURN_GET2'

        EXPORTING

          type   = sy-msgty

          cl     = sy-msgid

          number = sy-msgno

          par1   = sy-msgv1

          par2   = sy-msgv2

          par3   = sy-msgv3

          par4   = sy-msgv4

        IMPORTING

          return = es_return.

      EXIT.

    ENDIF.

* read all values

    CALL METHOD lr_variable->get_value

      EXPORTING

        i_user     = sy-uname

        i_restrict = ' '

      RECEIVING

        rto_value  = eto_varsel_all

      EXCEPTIONS

        error      = 1

        OTHERS     = 2.

    IF sy-subrc <> 0.

      e_subrc = 4.

      CALL FUNCTION 'BALW_BAPIRETURN_GET2'

        EXPORTING

          type   = sy-msgty

          cl     = sy-msgid

          number = sy-msgno

          par1   = sy-msgv1

          par2   = sy-msgv2

          par3   = sy-msgv3

          par4   = sy-msgv4

        IMPORTING

          return = es_return.

      EXIT.

    ENDIF.

* -- no error occured => store results to buffer

  IF i_buffer = 'X'.

    CLEAR ls_buffer.

    ls_buffer-area          = i_area.

    ls_buffer-variable      = i_variable.

    ls_buffer-subrc         = e_subrc.

    ls_buffer-s_return      = es_return.

    ls_buffer-type          = e_type.

    ls_buffer-to_varsel_all = eto_varsel_all.

    ls_buffer-to_varsel     = eto_varsel.

    ls_buffer-to_chanm      = eto_chanm.

    INSERT ls_buffer INTO TABLE st_buffer.
  ENDIF.
ENDFUNCTION.
4.2 Function module Z_SIMPLE_RELATION

FUNCTION z_simple_relation.

*"----------------------------------------------------------------------

*"*"Local interface:

*"  IMPORTING

*"     VALUE(I_AREA) TYPE  UPC_Y_AREA

*"     VALUE(I_VARIABLE) TYPE  UPC_Y_VARIABLE

*"     VALUE(I_CHANM) TYPE  UPC_Y_CHANM

*"     VALUE(ITO_CHANM) TYPE  UPC_YTO_CHA

*"  EXPORTING

*"     VALUE(ETO_CHARSEL) TYPE  UPC_YTO_CHARSEL

*"----------------------------------------------------------------------

* Change constant according to your needs

  CONSTANTS:

    l_source_var  TYPE upc_y_variable        VALUE 'SOURCE',

    l_source_area TYPE upc_y_area            VALUE '4PERF001',

    l_use_restricted_values TYPE boole-boole VALUE 'X',

    l_buffer_call TYPE boole-boole           VALUE 'X'.

  DATA:

      l_subrc      LIKE sy-subrc,

      ls_return    LIKE bapiret2,

      l_type       LIKE upc_var-vartype,

      lto_varsel_all TYPE upc_yto_charsel,

      lto_varsel   TYPE upc_yto_charsel,

      lto_var      TYPE upc_yto_charsel,

      lto_chanm    TYPE upc_yto_cha.

* read source value

  CALL FUNCTION 'Z_VARIABLE_GET_DETAIL'

    EXPORTING

      i_area         = l_source_area

      i_variable     = l_source_var

      i_buffer       = l_buffer_call

    IMPORTING

      e_subrc        = l_subrc

      es_return      = ls_return

      e_type         = l_type

      eto_varsel_all = lto_varsel_all

      eto_varsel     = lto_varsel

      eto_chanm      = lto_chanm.

  IF l_subrc <> 0.

    MESSAGE i136(upc_fw) WITH l_source_var.

*   Values of variable &1 cannot be determined

    EXIT.

  ENDIF.

* now, you are free to determine the logic on how to

* derive the values based on the source values.

* ----------------------------------------------------------------------

* Example 1: determine the next year

* Assumptions:

*  - we have only a single year in the selection

*  - the variable is of type characteristic

* ----------------------------------------------------------------------

  DATA: ls_varsel TYPE upc_ys_charsel,

        l_next_year(4) TYPE n,

        l_entries TYPE i.

* We have a single value for the year and only one characteristic

* => our value is stored in the first line

  IF l_use_restricted_values IS INITIAL.

    lto_var = lto_varsel_all.

  ELSE.

    lto_var = lto_varsel.

  ENDIF.

  READ TABLE lto_var INTO ls_varsel INDEX 1.

* check prerequisites:

* - record found?

  IF sy-subrc <> 0.

    MESSAGE i147(upc_fw) WITH l_source_var.

*   Variable &1 does not contain any values

    EXIT.

  ENDIF.

* - exactly one record and characteristic?

  DESCRIBE TABLE lto_var LINES l_entries.

  IF l_entries <> 1.

    MESSAGE i534(upc) WITH l_source_var.

*   Variable &1 must be restricted to a value

    EXIT.

  ENDIF.

* is there a next year?

  l_next_year = ls_varsel-low.

  IF l_next_year = 9999.

    MESSAGE i003(upc) WITH '9999 +1' '0FISCYEAR'.

*   Value &1 is not permitted for characteristic &2

    EXIT.

  ENDIF.

* find next year

  l_next_year = l_next_year + '0001'.

  ls_varsel-low = l_next_year.

* add to result table

  APPEND ls_varsel TO eto_charsel.

* ----------------------------------------------------------------

* End of Example 1

* ----------------------------------------------------------------

ENDFUNCTION.

4.3 Selection of an interval

FUNCTION z_selection_interval.

*"----------------------------------------------------------------------

*"*"Lokale Schnittstelle:

*"  IMPORTING

*"     VALUE(I_AREA) TYPE  UPC_Y_AREA

*"     VALUE(I_VARIABLE) TYPE  UPC_Y_VARIABLE

*"     VALUE(I_CHANM) TYPE  UPC_Y_CHANM

*"     VALUE(ITO_CHANM) TYPE  UPC_YTO_CHA

*"  EXPORTING

*"     VALUE(ETO_CHARSEL) TYPE  UPC_YTO_CHARSEL

*"----------------------------------------------------------------------

* Change constant according to your needs

  CONSTANTS:

    l_source_var1 TYPE upc_y_variable        VALUE 'SFROM',

    l_source_var2 TYPE upc_y_variable        VALUE 'STO',

    l_source_area TYPE upc_y_area            VALUE '4PERF001',

    l_use_restricted_values TYPE boole-boole VALUE 'X'.

  DATA:

      l_subrc      LIKE sy-subrc,

      ls_return    LIKE bapiret2,

      ls_interval   LIKE upc_ys_charsel,

      l_type       LIKE upc_var-vartype,

      lto_varsel_all TYPE upc_yto_charsel,

      lto_varsel   TYPE upc_yto_charsel,

      lto_var      TYPE upc_yto_charsel,

      lto_chanm    TYPE upc_yto_cha,

      ls_varsel TYPE upc_ys_charsel,

      l_entries TYPE i.

* read the content of the 'TO' variable.

  CALL FUNCTION 'Z_VARIABLE_GET_DETAIL'

    EXPORTING

      i_area         = l_source_area

      i_variable     = l_source_var2

    IMPORTING

      e_subrc        = l_subrc

      es_return      = ls_return

      e_type         = l_type

      eto_varsel_all = lto_varsel_all

      eto_varsel     = lto_varsel

      eto_chanm      = lto_chanm.

  IF l_subrc <> 0.

    MESSAGE i136(upc_fw) WITH l_source_var2.

*   Values of variable &1 cannot be determined

    EXIT.

  ENDIF.

* We assume a single value in the variable

* => our value is stored in the first line

  IF l_use_restricted_values IS INITIAL.

    lto_var = lto_varsel_all.

  ELSE.

    lto_var = lto_varsel.

  ENDIF.

  READ TABLE lto_var INTO ls_varsel INDEX 1.

* check prerequisites:

* - record found?

  IF sy-subrc <> 0.

    MESSAGE i147(upc_fw) WITH l_source_var2.

*   Variable &1 does not contain any values

    EXIT.

  ENDIF.

* - exactly one record and characteristic?

  DESCRIBE TABLE lto_var LINES l_entries.

  IF l_entries <> 1.

    MESSAGE i534(upc) WITH l_source_var2.

*   Variable &1 must be restricted to a value

    EXIT.

  ENDIF.

* store result of the ‘TO’ variable in the intervalselection

  ls_interval-high = ls_varsel-low.

* read the content of the 'FROM' variable

* this is mainly the identical Source code shown above

  CALL FUNCTION 'Z_VARIABLE_GET_DETAIL'

    EXPORTING

      i_area         = l_source_area

      i_variable     = l_source_var1

    IMPORTING

      e_subrc        = l_subrc

      es_return      = ls_return

      e_type         = l_type

      eto_varsel_all = lto_varsel_all

      eto_varsel     = lto_varsel

      eto_chanm      = lto_chanm.

  IF l_subrc <> 0.

    MESSAGE i136(upc_fw) WITH l_source_var1.

*   Values of variable &1 cannot be determined

    EXIT.

  ENDIF.

* We have a single value in the variable

* => our value is stored in the first line

  IF l_use_restricted_values IS INITIAL.

    lto_var = lto_varsel_all.

  ELSE.

    lto_var = lto_varsel.

  ENDIF.

  READ TABLE lto_var INTO ls_varsel INDEX 1.

* check prerequisites: - record found?

  IF sy-subrc <> 0.

    MESSAGE i147(upc_fw) WITH l_source_var1.

*   Variable &1 does not contain any values

    EXIT.

  ENDIF.

* - exactly one record and characteristic?

  DESCRIBE TABLE lto_var LINES l_entries.

  IF l_entries <> 1.

    MESSAGE i534(upc) WITH l_source_var1.

*   Variable &1 must be restricted to a value

    EXIT.

  ENDIF.

  ls_interval-low   = ls_varsel-low.

  ls_interval-opt   = 'BT'.

  ls_interval-sign  = 'I'.

  ls_interval-chanm = i_chanm.

  ls_interval-seqno = '0001'.

APPEND ls_interval TO eto_charsel.
ENDFUNCTION.

